
ABSTRACT. In this first paper, we consider how to design a slide to make a toboggan go from beginning to
end as quickly as possible.

1. INTRODUCTION

We express our optimization models in theAMPL modeling language [2]. This language provides a
common mechanism for conveying problems to codes to solve them. When solving problems we generally
use two different solvers: (a)LOQO [4, 5, 6, 1], which implements an interior-point method for general
nonlinear optimization and (b)SNOPT [3], which implements an active set strategy with a quasi-Newton
method for the QP subproblem.

2. TOBOGGANING

It wouldn’t be right to end our discussion of trajectory optimization without discussing the famous
Brachistochrone problem. We do it here in this last section in the context of designing a slide which will
get the rider from the beginning to the end in the shortest amount of time.

We assume in this study that the slide will be constructed out of frictionless material. We letx denote,
as usual, horizontal displacement and we lety denote vertical displacement. In contrast with earlier
conventions, we assume thaty increases as one moves downward. The slide will connect two points
having coordinates(x0, y0) = (0, 0) and(xf , yf ). The toboggan starts from rest at the top of the slide.
Hence, initially the toboggan has zero kinetic energy and zero potential energy (due to gravity). Since the

param n := 512;

param x {j in 0..n} := j/n;

# Variables
var y {j in 0..n} >= 0;

# Abbreviations
var dydx {j in 1..n}

= (y[j]-y[j-1])/(x[j]-x[j-1]);
var f {j in 1..n}

= sqrt( (1+dydx[j]ˆ2)/y[j-1] );

minimize time:
sum {j in 1..n}

f[j]*(x[j]-x[j-1]) ;

subject to y0: y[0] = 1.0e-12;
subject to yn: y[n] = 1;

subject to monotone {j in 1..n}:
dydx[j] >= 0;

let {j in 0..n} y[j] := x[j];

solve;

FIGURE 1. A working AMPL model for the Brachistochrone problem.

slide is frictionless, there is no loss of energy as heat. By conservation of energy, the total energy must
always remain zero. So, when the toboggan has dropped to levely, we have

1

2
mv2 −mgy = 0.

Here,v denotes the speed of the toboggan. From this relation we see that

v =
√

2gy.

Now, the time to do the run can be computed as an integral of differential chunks of time

T =

∫ T

0

dt =

∫ T

0

ds(t)

v(t)
.

This last integral can be reparametrized using any monotone function of time. The natural candidate is
horizontal displacementx. With this choice, we get

(1) T =

∫ xf

0

√
1 + y′(x)2√

2gy(x)
dx.

The problem then is to find a functiony(x) that minimizes this integral and satisfies the constraintsy(0) =
0 andy(xf ) = yf . Using calculus of variations, one can show that the general form of the solution to (1)
is a cycloid:

x = k2(θ − sin θ)

y = k2(1− cos θ).

However, to find the values ofk and θf to satisfy the original terminal conditions involves solving a
transcendental equation—not such an easy task.

The AMPL model expressing the minimization ofT as given in (1) is shown in Figure 1. It took some
tinkering before we were able to get to the working model shown in the Figure. The main issue is that
y(0) = 0 appears in the denominator of the integrand whenx = 0. Hence, the integral is a singular
integral. To address this, we first changed the boundary condition toy(0) = 10−12. Also, sincedydx[j]

param n := 512;

param y {j in 0..n} := (j/n);

# Variables
var x {j in 0..n};

# Abbreviations
var dxdy {j in 1..n}

= (x[j] - x[j-1])/(y[j] - y[j-1]);
var f {j in 1..n}

= sqrt( (dxdy[j]ˆ2+1)/((y[j]+y[j-
1])/2) );

minimize time:
sum {j in 1..n} f[j]*(y[j]-y[j-1]) ;

subject to x0: x[0] = 0;
subject to xn: x[n] = 1;

let {j in 0..n} x[j] := y[j];

solve;

FIGURE 2. A second workingAMPL model for the Brachistochrone problem.

appearing in the definition off[j] represents the value of the derivative at the midpoint of the interval
[j − 1, j], one would expect to use the best estimate fory at this same place, i.e.,(y[j]+y[j-1])/2 .
However, with this choice for denominator in the integrand, the optimal solution exhibits a very large
jump discontinuity atx = 0. Presumably this is caused by the singularity but the details elude us. Using
y[j-1] , as shown, works but changing it toy[j] renders the problem unsolvable to bothLOQO and
SNOPT. Again, the reason remains a mystery. It is easy to think of lots of other things to try (and we did)
but we stop here in favor of a different line of attack. The model shown in the Figure is solved byLOQO

in 26 iterations. It takes 0.85 seconds on a 366 MHz PC.
Instead of using horizontal displacement as the parameterization variable in (1), we could equally well

have chosen to use the vertical displacement variable. With this choice, we get

(2) T =

∫ yf

0

√
1 + x′(y)2

√
2gy

dy.

TheAMPL model for this formulation of the problem is shown in Figure 2.LOQO solves this model in just
10 iterations. It takes only 0.26 seconds on a 366 MHz PC. Clearly, from a numerical perspective, this
formulation is much better than the previous one.

2.1. Lesson. The lesson to take away from this case study is that one should consider a variety of ways
to formulate a given problem. Some might be much easier to solve than others.

